Key Homework 22

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

Initialization

```
> restart;
```

4: Page 227

Code the differential equation and the corresponding homogeneous equation.

```
> deq:=diff(y(t), t$2)+y(t)=5*cos(t);
  deqh:=lhs(deq)=0;
```

$$deq := \left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) + \mathrm{y}(t) = 5\cos(t)$$

$$deqh := \left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) + \mathrm{y}(t) = 0$$

Find the eigenvalues and construct the general solution of the homogeneous equation.

```
> aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
  evals:=solve(aux, r);
```

$$aux := r^2 + 1 = 0$$

$$evals := I, -I$$

```
> Y[1]:=cos(t);
  Y[2]:=sin(t);
  ygh:=add(c[k]*Y[k], k=1..2);
```

$$Y_1 := \cos(t)$$

$$Y_2 := \sin(t)$$

$$ygh := c_1 \cos(t) + c_2 \sin(t)$$

Use the method of undetermined coefficients to find a particular solution for the inhomogeneous equation. Since $\cos(t)$ is a solution for the homogeneous equation, we use a trial solution of the form.

```
> ytry:=t*(a*cos(t)+b*sin(t));
```

$$ytry := t\,(a\cos(t) + b\sin(t))$$

Find the parameters *a* and *b* by substitution of the trial solution into the inhomogeneous equation.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  val_p:=solve(identity(eq_p, t), {a,b});
```

$$eq\_p := -2\,a\sin(t) + 2\,b\cos(t) + t\,(-a\cos(t) - b\sin(t)) + t\,(a\cos(t) + b\sin(t)) = 5\cos(t)$$

$$val\_p := \{a = 0, b = \frac{5}{2}\}$$

A particular solution of the inhomogeneous equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := \frac{5}{2}\,t\sin(t)$$

and the general solution of the inhomogeneous equation takes the form

```
> yg:=ygh+yp;
```

$$yg := c_1\cos(t) + c_2\sin(t) + \frac{5}{2}\,t\sin(t)$$

Implement the initial conditions.

```
> eq1:=eval(subs(t=0, yg)=0);
  eq2:=eval(subs(t=0, diff(yg, t))=1);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 = 0$$

$$eq2 := c_2 = 1$$

$$val\_c := \{c_1 = 0, c_2 = 1\}$$

```
> sol:=subs(val_c, yg);
```

$$sol := \sin(t) + \frac{5}{2} t \sin(t)$$

Finally, we plot the solution.

```
> plot(sol, t=0..100);
```




```
>
```

5: Page 227

Since the exercise is phrased in terms of $\omega = \sqrt{\frac{k}{m}}$, we divide both sides of the differential equation

$$m \left(\frac{d^2}{dt^2} y(t) \right) + k\, y(t) = F_0 \cos(\gamma\, t)$$

by m and obtain

```
> deq:=diff(y(t), t$2)+omega^2*y(t)=F[0]/m*cos(gamma*t);
```

$$deq := \left(\frac{d^2}{dt^2} y(t) \right) + \omega^2\, y(t) = \frac{F_0 \cos(\gamma\, t)}{m}$$

Tell the system that m and ω are both positive.

```
> assume(m>0);
  assume(omega>0);
```

5a

Code the corresponding homogeneous equation.

```
> deqh:=lhs(deq)=0;
```

$$deqh := \left(\frac{d^2}{dt^2} y(t) \right) + \omega^2\, y(t) = 0$$

Find the eigenvalues and construct the general solution of the homogeneous equation.

```
> aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
  evals:=solve(aux, r);
```

$$aux := r^2 + \omega^2 = 0$$

$$evals := \omega\, I, -I\, \omega$$

```
> Y[1]:=cos(Im(evals[1])*t);
  Y[2]:=sin(Im(evals[1])*t);
  ygh:=add(c[k]*Y[k], k=1..2);
```

$$Y_1 := \cos(\omega t)$$

$$Y_2 := \sin(\omega t)$$

$$ygh := c_1 \cos(\omega t) + c_2 \sin(\omega t)$$

Use the method of undetermined coefficients to find a particular solution for the inhomogeneous equation. Since $\cos(\gamma t)$ is assumed to not be a solution for the homogeneous equation, we use a trial solution of the form.

```
> ytry:=a*cos(gamma*t)+b*sin(gamma*t);
```

$$ytry := a \cos(\gamma t) + b \sin(\gamma t)$$

Find the parameters *a* and *b* by substitution of the trial solution into the inhomogeneous equation.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  val_p:=solve(identity(eq_p, t), {a,b});
```

$$eq\_p := -a \cos(\gamma t) \gamma^2 - b \sin(\gamma t) \gamma^2 + \omega^2 (a \cos(\gamma t) + b \sin(\gamma t)) = \frac{F_0 \cos(\gamma t)}{m}$$

$$val\_p := \left\{ b = 0, a = \frac{F_0}{m(-\gamma^2 + \omega^2)} \right\}$$

A particular solution of the inhomogeneous equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := \frac{F_0 \cos(\gamma t)}{m(-\gamma^2 + \omega^2)}$$

and the general solution of the inhomogeneous equation takes the form

```
> yg:=ygh+yp;
```

$$yg := c_1 \cos(\omega t) + c_2 \sin(\omega t) + \frac{F_0 \cos(\gamma t)}{m(-\gamma^2 + \omega^2)}$$

Implement the initial conditions.

```
> eq1:=eval(subs(t=0, yg)=0);
  eq2:=eval(subs(t=0, diff(yg, t))=0);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + \frac{F_0}{m(-\gamma^2 + \omega^2)} = 0$$

$$eq2 := c_2 \omega = 0$$

$$val\_c := \left\{ c_2 = 0, c_1 = -\frac{F_0}{m(-\gamma^2 + \omega^2)} \right\}$$

The equation of motion for the system is given by

```
> sol:=y(t)=subs(val_c, yg);
```

$$sol := y(t) = -\frac{F_0 \cos(\omega t)}{m(-\gamma^2 + \omega^2)} + \frac{F_0 \cos(\gamma t)}{m(-\gamma^2 + \omega^2)}$$

```
>
```

5b

Using the trigonometric identity

$$\cos(p) - \cos(q) = -2 \sin\left(\frac{p+q}{2}\right) \sin\left(\frac{p-q}{2}\right)$$

we obtain that

$$\frac{F_0 (\cos(\gamma t) - \cos(\omega t))}{m(\omega^2 - \gamma^2)} = \frac{2 F_0 \sin\left(\frac{(\omega + \gamma) t}{2}\right) \sin\left(\frac{(\omega - \gamma) t}{2}\right)}{m(\omega^2 - \gamma^2)}$$

5c

```
> pars:={F[0]=32, m=2, omega=9, gamma=7};
```

$$pars := \{ F_0 = 32, \gamma = 7, m = 2, \omega = 9 \}$$

```
> ex:=subs(pars, rhs(sol));
```

$$ex := -\frac{1}{2}\cos(9\,t) + \frac{1}{2}\cos(7\,t)$$

In compact form, using the identity found in Part (b), this translates into

```
> yc:=A*sin(t)*sin(8*t);
  pars_yc:=solve(identity(ex=yc, t), {A});
  exc:=subs(pars_yc, yc);
```

$$yc := A\sin(t)\sin(8\,t)$$
$$pars\_yc := \{A = 1\}$$
$$exc := \sin(t)\sin(8\,t)$$

Plot the solution to reveal the beats.

```
> plot(ex, t=0..10);
```



The beats become more clearly visible if we add the low frequency oscillation to the picture.

```
> plot([ex, sin(t), -sin(t)], t=0..10, color=[red, blue, blue]);
```




```
>
```

8: Page 228

Code the paramaters, the differential equation and the corresponding homogeneous equation.

```
> pars:={m=2, b=8, k=6, F[0]=18, gamma=0};
  deq_general:=m*diff(y(t), t$2)+b*diff(y(t), t)+k*y(t)=F[0]*cos(gamma*t);
  deq:=eval(subs(pars, deq_general));
  deqh:=lhs(deq)=0;
```

$$pars := \{\gamma = 0, m = 2, b = 8, k = 6, F_0 = 18\}$$
$$deq\_general := m\left(\frac{d^2}{dt^2}y(t)\right) + b\left(\frac{d}{dt}y(t)\right) + k\,y(t) = F_0\cos(\gamma\,t)$$
$$deq := 2\left(\frac{d^2}{dt^2}y(t)\right) + 8\left(\frac{d}{dt}y(t)\right) + 6\,y(t) = 18$$
$$deqh := 2\left(\frac{d^2}{dt^2}y(t)\right) + 8\left(\frac{d}{dt}y(t)\right) + 6\,y(t) = 0$$

Find the eigenvalues and construct the general solution of the homogeneous equation.

```
> aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
  evals:=solve(aux, r);
```

$$aux := 2\,r^2 + 8\,r + 6 = 0$$
$$evals := -1, -3$$

```
> ygh:=add(c[k]*exp(evals[k]*t), k=1..2);
```

$$ygh := c_1\,e^{(-t)} + c_2\,e^{(-3\,t)}$$

Use the method of undetermined coefficients to find a particular solution for the inhomogeneous equation. Since 18 is not a solution for the homogeneous equation, we use a trial solution of the form.

```
> ytry:=C;
```

$$ytry := C$$

Find the parameter C by substituting the trial solution into the inhomogeneous equation.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  val_p:=solve(identity(eq_p, t), {C});
```

$$eq\_p := 6\,C = 18$$

$$val\_p := \{C = 3\}$$

A particular solution of the inhomogeneous equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := 3$$

and the general solution of the inhomogeneous equation takes the form

```
> yg:=ygh+yp;
```

$$yg := c_1\,\mathbf{e}^{(-t)} + c_2\,\mathbf{e}^{(-3\,t)} + 3$$

Implement the initial conditions.

```
> eq1:=eval(subs(t=0, yg)=0);
  eq2:=eval(subs(t=0, diff(yg, t))=0);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + c_2 + 3 = 0$$

$$eq2 := -c_1 - 3\,c_2 = 0$$

$$val\_c := \{c_1 = \frac{-9}{2}, c_2 = \frac{3}{2}\}$$

```
> sol:=subs(val_c, yg);
```

$$sol := -\frac{9}{2}\,\mathbf{e}^{(-t)} + \frac{3}{2}\,\mathbf{e}^{(-3\,t)} + 3$$

Plot the solution.

```
> plot(sol, t=0..20);
```



The spring is stretched, or compressed, by a constant force. Hence, as t tends to infinity, its length will be enlarged, or shortened, by a constant amount.

```
> L:=Limit(sol, t=infinity)=limit(sol,t=infinity);
```

$$L := \lim_{t \to \infty} -\frac{9}{2}\,\mathbf{e}^{(-t)} + \frac{3}{2}\,\mathbf{e}^{(-3\,t)} + 3 = 3$$

Observe that this result is commensurate with the differential equation, because

$$\frac{F_0}{k} = \frac{18}{6} = 3$$

11: Page 228

Code the paramaters, the differential equation and the corresponding homogeneous equation.

```
> pars:={m=8/32, b=1, k=10, F[0]=2, gamma=2};
  deq_general:=m*diff(y(t), t$2)+b*diff(y(t), t)+k*y(t)=F[0]*cos(gamma*t);
```

```
deq:=eval(subs(pars, deq_general));
deqh:=lhs(deq)=0;
```

$$pars := \{b = 1, k = 10, F_0 = 2, \gamma = 2, m = \frac{1}{4}\}$$

$$deq\_general := m\left(\frac{d^2}{dt^2}\,y(t)\right) + b\left(\frac{d}{dt}\,y(t)\right) + k\,y(t) = F_0\cos(\gamma\,t)$$

$$deq := \frac{1}{4}\left(\frac{d^2}{dt^2}\,y(t)\right) + \left(\frac{d}{dt}\,y(t)\right) + 10\,y(t) = 2\cos(2\,t)$$

$$deqh := \frac{1}{4}\left(\frac{d^2}{dt^2}\,y(t)\right) + \left(\frac{d}{dt}\,y(t)\right) + 10\,y(t) = 0$$

Find the eigenvalues and construct the general solution of the homogeneous equation.

```
> aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
  evals:=solve(aux, r);
```

$$aux := \frac{1}{4}r^2 + r + 10 = 0$$

$$evals := -2 + 6\,I, -2 - 6\,I$$

```
> ygh:=c[1]*exp(Re(evals[1])*t)*cos(Im(evals[1])*t)+c[2]*exp(Re(evals[1])*t)*sin(Im(eval
  s[1])*t);
```

$$ygh := c_1\,\mathbf{e}^{(-2\,t)}\cos(6\,t) + c_2\,\mathbf{e}^{(-2\,t)}\sin(6\,t)$$

Use the method of undetermined coefficients to find a particular solution for the inhomogeneous equation. Since $\cos(2\,t)$ is not a solution for the homogeneous equation, we use a trial solution of the form.

```
> ytry:=a[1]*cos(2*t)+a[2]*sin(2*t);
```

$$ytry := a_1\cos(2\,t) + a_2\sin(2\,t)$$

Find the parameters a_1 and a_2 by substituting the trial solution into the inhomogeneous equation.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  val_p:=solve(identity(eq_p, t), {a[1], a[2]});
```

$$eq\_p := 9\,a_1\cos(2\,t) + 9\,a_2\sin(2\,t) - 2\,a_1\sin(2\,t) + 2\,a_2\cos(2\,t) = 2\cos(2\,t)$$

$$val\_p := \{a_1 = \frac{18}{85}, a_2 = \frac{4}{85}\}$$

A particular solution of the inhomogeneous equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := \frac{18}{85}\cos(2\,t) + \frac{4}{85}\sin(2\,t)$$

With a little bit of effort, we can write this in the compact form $A\sin(2\,t + \theta)$.

```
> ypc:=A*sin(2*t+theta);
  pars_yp:=solve(identity(yp=ypc, t), {A, theta});
```

$$ypc := A\sin(2\,t + \theta)$$

$$pars\_yp := \{A = \frac{2\sqrt{85}}{85}, \theta = \arctan\left(\frac{9}{2}\right)\}, \{A = -\frac{2\sqrt{85}}{85}, \theta = \arctan\left(\frac{9}{2}\right) - \pi\}$$

```
> yp:=subs(pars_yp[1], ypc);
```

$$yp := \frac{2}{85}\sqrt{85}\,\sin\left(2\,t + \arctan\left(\frac{9}{2}\right)\right)$$

and the general solution of the inhomogeneous equation takes the form

```
> yg:=ygh+yp;
```

$$yg := c_1\,\mathbf{e}^{(-2\,t)}\cos(6\,t) + c_2\,\mathbf{e}^{(-2\,t)}\sin(6\,t) + \frac{2}{85}\sqrt{85}\,\sin\left(2\,t + \arctan\left(\frac{9}{2}\right)\right)$$

Implement the initial conditions.

```
> eq1:=eval(subs(t=0, yg)=0);
  eq2:=eval(subs(t=0, diff(yg, t))=0);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + \frac{18}{85} = 0$$

$$eq2 := -2\,c_1 + \frac{8}{85} + 6\,c_2 = 0$$

$$val\_c := \{c_2 = \frac{-22}{255}, c_1 = \frac{-18}{85}\}$$

```
> sol:=subs(val_c, yg);
```

$$sol := -\frac{18}{85}\,\mathbf{e}^{(-2\,t)}\cos(6\,t) - \frac{22}{255}\,\mathbf{e}^{(-2\,t)}\sin(6\,t) + \frac{2}{85}\sqrt{85}\,\sin\left(2\,t + \arctan\left(\frac{9}{2}\right)\right)$$

The resonance frequency for the system is given by

$$\frac{\gamma_r}{2\,\pi} = \frac{\sqrt{\frac{k}{m} - \frac{b^2}{2\,m^2}}}{2\,\pi}$$

This result was derived in class and is equivalent to formula 15 on Page 223 of the textbook.

```
> resonance_frequency:=gamma[r]/(2*Pi)=simplify(subs(pars,
  sqrt(k/m-b^2/(2*m^2)))/(2*Pi));
```

$$resonance\_frequency := \frac{1}{2}\,\frac{\gamma_r}{\pi} = \frac{2\sqrt{2}}{\pi}$$

```
>
```